Stringvergleiche in FoxPro

Ein Kapitel über Stringvergleiche? Den Programmierern, die andere Sprachen verwenden, käme das sicherlich nicht in den Sinn, schließlich haben die meisten Sprachen nur eine Funktion, um festzustellen, ob zwei Strings gleich sind oder nicht. Und nicht nur das. Diese Funktion verhält sich auch noch immer gleich! FoxPro war schon immer ein wenig anders, und hier ist der Stringvergleich eine ziemlich komplizierte Angelegenheit. Nicht nur, daß es zwei Befehle gibt, nämlich "=" und "= =", diese verhalten sich auch noch unterschiedlich, je nachdem wie einige SET-Befehle gesetzt sind. In diesem Kapitel will ich einmal darauf eingehen, wie der Stringvergleich in FoxPro funktioniert, wovon er abhängig ist und wie zahlreiche Funktionen auf die unterschiedlichen Einstellungen reagieren. 

Der = Operator

Der am häufigsten zum Stringvergleich genutzte Operator ist "=". In der Standardeinstellung mit SET EXACT OFF vergleicht er Strings nur bis zur Länge des auf der rechten Seite stehenden String. Ist also diese kürzer als der linke, so werden dennoch beide Strings als identisch angesehen. Andersherum ist das nicht gültig. Wenn der rechte String länger ist als der linke, sind beide Strings unterschiedlich: 

? "String, der lang ist" = "String" && .T.
? "String" = "String, der lang ist" && .F. 
Der == Operator

Dagegen müssen bei "==" die Strings nicht nur identisch sein, sondern auch gleich lang sein. Der Stringvergleich mit "==" nennt sich daher auch der "exakte Stringvergleich". "==" ist übrigens unabhängig von SET EXACT und diversen anderen Einstellungen. Wenn Sie also einen Stringvergleich brauchen, der unabhängig von diversen äußeren Einflüssen seinen Dienst tun soll, nehmen Sie "==" als Operator. Der angenehme Nebeneffekt ist hier, daß "==" bei Speichervariablen sogar etwas schneller als "=" ist. Das folgende Beispiel illustriert den Gebrauch des Operators für den exakten Stringvergleich: 

? "String" == "String"  && .T. 
? "String1" == "String" && .F. 
SET EXACT

Welche Rolle spielt nun SET EXACT in der ganzen Geschichte? SET EXACT kann zwei Werte annehmen, die da lauten ON und OFF. Die Standardeinstellung ist OFF, mit der sich Stringvergleiche wie oben beschrieben verhalten. Mit ON schalten sie den exakten Stringvergleich ein. Oftmals wird angenommen, daß "=" sich bei SET EXACT ON genauso verhält, wie "==". Dieser Eindruck entsteht leicht durch die Namengebung des SET EXACT-Befehls. Auf den ersten Blick scheint sich dieses Verhalten auch zu bestätigen: 

? "String" = "String"        && .T. 
? "String1" = "String"  && .F. 
Dies liefert die gleichen Ergebnisse, wie oben bereits "==" lieferte. Aber Vorsicht! Der "=" Operator ignoriert grundsätzlich alle Leerzeichen, die den Strings folgen: 

? "String " = "String"   && .T. 
? "String" = "String "   && .T. 
? "String " == "String"  && .F. 
? "String" == "String "  && .F. 
Der Grund für dieses Verhalten ist schnell erklärt. Wäre es nämlich anders, könnten Sie Tabellenfelder nicht mehr mit SET EXACT ON vergleichen, ohne zuvor ALLTRIM() verwenden zu müssen. So aber können Sie auch Tabellenfelder vergleichen, die ja bekanntermaßen immer mit Leerzeichen auf die definierte Länge aufgefüllt werden. Wenn Sie also mit SET EXACT ON und "=" arbeiten, können sie sich das so merken, daß auf beide Strings erst die Funktion RTRIM() angewandt wird, bevor die Strings verglichen werden. Die folgenden Zeilen sind also identisch: 

SET EXACT ON 
? String1 = String2 
? RTRIM(String1) = RTRIM(String2) 
Mit diesem Wissen können Sie nun auch erklären, warum Sie einen String in FoxPro nie mit einem Leerstring vergleichen sollten. Wenn SET EXACT OFF gesetzt wurde, liefert ein solcher Vergleich nämlich immer .T. zurück: 

? "String" = ""         && .T. 

Wenn Sie diesen Befehl umdrehen, so liefert FoxPro wie erwartet ein .F. zurück: 

? "" = "String"         && .F. 
Nicht also nur SET EXACT ON ist beim Stringvergleich entscheidend, auch die Reihenfolge ist von Bedeutung. Dies ist großer Unterschied zu vielen anderen Sprachen, bei denen die Operanden in beliebiger Reihenfolge im Vergleichsoperator verwendet werden können. Wiederum verhält sich der Vergleich mit einem Leerstring anders, wenn Sie SET EXACT ON setzen: 

? "String" = ""         && .F. 
Auch wenn Sie hier scheinbar einen String mit einem Leerstring vergleichen können, hüten Sie sich davor, denn wieder werden Leerzeichen ignoriert: 

? "   " = ""            && .T. 
EMPTY()

Wenn Sie also wissen wollen, ob ein String wirklich leer ist, verwenden Sie den exakten Stringvergleich mit "==" oder prüfen sie, ob die Länge 0 ist, indem Sie die Funktion LEN() aufrufen. Warum empfehle ich hier nicht die Funktion EMPTY(), die ja offenbar für den Test auf einen Leerstring gedacht ist? Nun, wann ist denn ein String überhaupt leer?  Ein String ist zum Beispiel dann leer, wenn er gar nichts enthält, also "" entspricht. Er ist aber auch dann leer, wenn er Leerzeichen enthält, das wußten Sie vermutlich schon. Was aber oftmals unbekannt ist, ist, daß ein String auch leer ist, wenn er einen der folgenden Codes enthält:  

CHR(9)   = Tabulator, 
CHR(13) = Carriage Return und 
Chr(10)  = Line feed. 
Der folgende Ausdruck liefert deswegen .T. zurück: 

? EMPTY( " "+Chr(9)+Chr(13)+Chr(10) ) 
Wenn Sie sicher gehen wollen, daß ein Ausdruck nur Leerzeichen oder nichts enthält, müssen Sie LEN(ALLTRIM(String))=0 als Bedingung verwenden. Denn ALLTRIM() entfernt nur Leerzeichen, nicht aber die anderen Codes, die ebenfalls von EMPTY() ignoriert werden. 

SET COLLATE

Glauben Sie nun nicht, wir wären am Ende des Artikels angelangt. SET EXACT ist nur einer der Befehle, der den Stringvergleich beeinflußt. Für den deutschsprachigen Raum ist SET COLLATE ein weiterer sehr wichtiger Befehl. Laut Hilfe bestimmt dieser Befehl die Reihenfolge bei Indizier- und Sortiervorgängen. Sein Einfluß auf den Stringvergleich ist daraus nicht offensichtlich. Schauen Sie sich einmal nachstehendes Programm an: 

SET COLLATE TO "MACHINE" 
? "String" = "string"   && .F. 
SET COLLATE TO "GENERAL" 
? "String" = "string"   && .T. 
Der Einfluß von SET COLLATE beschränkt sich also mitnichten nur auf Indizes oder auf Tabellenfelder! Auch jeder ganz normale String ist davon betroffen. Alle Vergleiche werden immer nach der aktuellen COLLATE-Einstellung durchgeführt. Dieses Verhalten ist durchaus verständlich, denn schließlich wollen Sie die Tabellenfelder auch dann noch nach dem gleichen Schema wie im Index vergleichen können, nachdem Sie dieses Feld in eine Speichervariable eingelesen haben. Nichtsdestotrotz lauert hier eine böse Falle auf Sie. Diese tritt immer dann in Erscheinung, wenn Sie Daten vergleichen wollen, die eigentlich kein Text sind: 

? Chr(1) = Chr(2) && .F. 
SET COLLATE TO "GENERAL" 
? Chr(1) = Chr(2) && .T. 

Mit GENERAL sind also binäre Zeichen, die keine darstellbaren Buchstaben sind, identisch. Keine Regel ohne Ausnahme, denn auch das gilt nicht für alle Zeichen. Ausgenommen davon ist CHR(9), der Tabulator: 

? Chr(1) = Chr(9) && .F. 
Wozu brauche ich binäre Zeichen, werden Sie sich vielleicht jetzt fragen? Nun, viele Programmierer benutzen binärcodierte Strings als Primärschlüssel in Tabellen. Zu Zeiten von FoxPro 2.x wurden binäre Zeichen in Strings oft vermieden, da es bekanntermaßen Konvertierungsprobleme durch unterschiedliche Codepages und ähnliches gab. Aber seit Visual FoxPro gibt es ja den neuen Feldtyp Character (Binary). VFP unterzieht Felder dieses Typs keiner irgendwie gearteten Konvertierung und so kann man nun wunderbar alle 256 möglichen Zeichen für Primärschlüssel verwenden. 

Leider ist das nur die halbe Wahrheit, denn wenn SET COLLATE auf einen anderen Wert als MACHINE gesetzt ist, unterliegt auch dieses Feld der Sortierfolge. 

SET COLLATE TO "GENERAL" 
CREATE TABLE Test( Feld CB(4) ) 
INSERT INTO Test (Feld) VALUE (CHR(1)+"ABC") 
? Test.Feld = CHR(2)+"ABC" 
Und siehe da, trotz binärem Feld sind beide Werte identisch. Sollten Sie also binäre Zeichen irgendwie verwenden, so stellen Sie sicher, daß SET COLLATE auf MACHINE gesetzt wird, während Sie Vergleiche mit diesen Feldern durchführen. Gerne würde ich Ihnen empfehlen, nur "==" für binäre Stringvergleiche zu benutzen, aber leider ist auch hier wieder einmal eine Falle verborgen. Lassen Sie mich noch einmal auf das obige Beispiel zurückkommen und diesmal den exakten Stringvergleich mittels "==" durchführen: 

? test.Feld == CHR(2)+"ABC"        && .F. 
Wunderbar! Genau dieses Ergebnis wollten wir haben. Und nun prüfen wir das genaue Gegenteil, nämlich ob die Strings vielleicht unterschiedlich sind: 

? test.Feld # CHR(2)+"ABC"         && .F. 
Hmm... Zwei Strings, die weder gleich noch ungleich sind? Leider gibt es in FoxPro keinen Befehl, der eine exakte Ungleichheit ermittelt. "#" (und auch "<>" und "!=") arbeitet genauso wie "=", wird also sowohl von SET EXACT als auch SET COLLATE beeinflußt. Wenn Sie die Ungleichheit von binären Strings ermitteln wollen, verwenden Sie immer folgende Konstruktion: 

? NOT (test.Feld == CHR(2)+"ABC")        && .T. 
Weitere Funktionen

Da stellt sich natürlich nun die Frage, welche anderen Befehle und Funktionen von diesen Einstellungen auch noch betroffen sind. Da wäre natürlich wie eben erwähnt der Ungleichheitsoperator "#". Dieser liefert grundsätzlich das genaue Gegenteil von "=" zurück. Verwenden Sie also nicht "#", wenn Sie eigentlich "NOT ==" meinen. Das sind zwei ganz unterschiedliche Paar Schuhe. 

Dann gibt es noch die Befehle größer als, kleiner als, größer oder gleich wie und kleiner oder gleich wie. Auch diese Befehle verhalten sich ebenso wie "=", sind also abhängig von SET COLLATE und SET EXACT. Die einfachste Aussage zu ihnen ist, immer wenn "=" wahr zurückgibt, liefern diese Operatoren ein logisches falsch zurück. Also ist beispielsweise "String  " > "String" immer falsch, da nachfolgende Leerzeichen ja ignoriert werden und somit die beiden Strings identisch sind. Schwieriger dagegen ist das Verhalten einzuschätzen, das von SET EXACT abhängig ist: 

SET EXACT ON
? "String1" > "String"  && .T. 
SET EXACT OFF 
? "String1" > "String"  && .F. 
Größer als (oder auch größer gleich) verhält sich hier also wie "=". Da im unteren Fall beide Strings als identisch angesehen werden, kann nicht einer von ihnen größer sein. Kleiner als verhält sich entsprechend. 

SET EXACT ON 
? "String1" < "String"  && .F. 
SET EXACT OFF 
? "String1" < "String"  && .F. 

Der obere Ausdruck liefert .F., da "String1" ja größer ist, der untere dagegen liefert .F., da beide Strings beim Vergleich mit SET EXACT OFF identisch sind. Schwieriger wird das Verhalten der Operatoren wieder einmal, wenn Leerzeichen ins Spiel kommen. Gehen wir da einmal systematisch vor. Wenn der linke Ausdruck mehr Leerzeichen enthält, aber ansonsten vollkommen identisch ist, so liefern alle Ausdrücke mit allen SET EXACT-Einstellungen immer .F.: 

SET EXACT OFF 
? "String " < "String"  && .F. 
? "String " > "String"  && .F. 
SET EXACT ON 
? "String " < "String"  && .F. 
? "String " > "String"  && .F. 
Bis auf die letzte Zeile würden wird das so erwarten. Ein mit Leerzeichen verlängerter String kann ja kaum kleiner sein. Größer kann er bei SET EXACT OFF auch nicht sein, da ja beide Strings identisch sind. Aber die letzte Zeile ist wieder eine typische Falle. Trotz SET EXACT ON werden alle nachfolgenden Leerzeichen auf der linken Seite ignoriert, wie wir bereits am Anfang dieses Artikels festgestellt haben. Da also beide Strings gleich sind, kann der eine eben nicht größer sein.  Nun wollen wir einmal die gleiche Konstellation betrachten, diesmal allerdings mit dem Leerzeichen auf der rechten Seite. 

SET EXACT OFF 
? "String" < "String "  && .T. 
? "String" > "String "  && .F. 
SET EXACT ON 
? "String" < "String "  && .F. 
? "String" > "String "  && .F. 
Wieder kann der String ohne Leerzeichen natürlich nicht größer sein, als der mit Leerzeichen. Auch die erste Aussage erscheint logisch, da EXACT OFF ja Strings immer nur auf die Länge des rechten Strings vergleicht. Somit ist "String " in der Tat größer als "String". Aber die dritte Zeile macht wieder Probleme, denn würden wir hier nicht auch ein .T. erwarten?  Es handelt sich doch um einen exakten Vergleich. Wie wir bereits am Anfang dieses Artikels festgestellt haben, wird bei SET EXACT ON ein implizites RTRIM() auf beide Strings ausgeführt. Wie sie dadurch leicht feststellen werden, sind "String" und "String " also bei SET EXACT ON immer identisch, gleichgültig, in welcher Reihenfolge sie stehen. Deswegen wird ">" oder "<" immer .F. zurückliefern, wenn sich beide Strings nur in der Anzahl der Leerzeichen am Ende unterscheiden. 

Offensichtlich fehlt uns hier die Möglichkeit des exakten Stringvergleiches im Sinne von SET EXACT ON. Wie können wir feststellen, daß "String " größer als "String" ist? Für FoxPro sind solche Strings identisch, auch wenn SET EXACT auf ON gesetzt wurde. Also prüfen wir in diesem Fall als erstes, ob die Strings identisch sind. Wenn ja, prüfen wir nach, ob der linke String länger als der zweite ist. In diesem Fall muß der linke String größer als der rechte sein, da sie sich ja nur um eine Anzahl von Leerzeichen unterscheiden können: 

PROCEDURE IstGroesser 
PARAMETER cStr1, cStr2 
IF SET("EXACT") = "ON"
  IF cStr1 = cStr2
     RETURN LEN(cStr1) > LEN(cStr2 )
  ELSE
     RETURN cStr1 > cStr2
  ENDIF 
ELSE
  RETURN cStr1 > cStr2 
ENDIF 
Weitaus schwieriger ist ein exakter Vergleich im Sinne von "==". Hier können wir nicht auf ">" oder "<" zurückgreifen, da diese ja wie auch "=" von SET COLLATE abhängig sind. Um das zu umgehen, prüfen wir erst auf exakte Gleichheit mittels "==". Sollten Sie unterschiedlich sein, so ist einer der Strings größer als der andere, es gilt nur herauszufinden welcher. Dazu vergleichen wir anschließend die Länge. Der längere String ist logischerweise größer, als der kürzere. Wenn beide Strings gleich lang sind, bleibt uns keine andere Möglichkeit, als den ASCII Code jedes Zeichen des einen Strings mit dem ASCII Code des entsprechenden Zeichen im anderen String zu vergleichen. 

Diese Lösung führt uns auch gleich weiter zum nächsten Problem. Wenn Sie Buchstaben vergleichen, machen Sie sich klar, ob Sie die Buchstaben oder deren ASCII-Code vergleichen wollen. Mit SET COLLATE TO "MACHINE" besteht da kein Unterschied, wohl aber bei allen anderen Sortierfolgen: 

SET COLLATE TO "GENERAL" 
? "s" = "S"            && .T. 
? ASC("s") = ASC("S")  && .F. 
Noch offensichtlicher wird dies, wenn ein Zeichen, mehreren anderen entspricht. Die deutsche Sprache hat ein solches Zeichen beispielsweise in Form des "ß": 

? "ß" = "ss"   && .T. 
Wie wollen Sie jetzt den ASCII-Code von "ß" mit dem ASCII Code zweier Zeichen vergleichen? Logische Zeichen, wie sie FoxPro in Sortierfolgen verwendet, haben also keinen ASCII-Code. Natürlich gibt es keinen Fehler, wenn Sie den ASCII-Code ermitteln wollen, machen Sie sich aber klar, daß das Ergebnis nur beschränkt brauchbar ist. Auf keinen Fall ist es eine adäquate Repräsentation des Zeichens. 

Und es gibt noch mehr Funktionen in FoxPro, die sich mit dem Stringvergleich beschäftigen. Diese kann man in zwei Gruppen einteilen: a) Funktionen, die ganze Strings vergleichen und b) Funktionen, die Teilstrings vergleichen. Die Funktionen der ersten Gruppe gehorchen den Gesetzen von "=", die der zweiten Gruppe denen von "==". 

Ein wichtige Funktion der ersten Gruppe ist INLIST(). Sie können sich eine INLIST() Zeile immer als eine Reihe von einzelnen Stringvergleichen vorstellen. So ist zum Beispiel 

INLIST(Var,"String1","String2") 
identisch mit: 
Var="String1" OR Var="String2" 
Damit ergibt sich auch automatisch, wie Strings verglichen werden, inklusive aller Fallen, die wir beim normalen Stringvergleich gefunden haben. Ein extremes Beispiel dafür ist 

SET EXACT ON 
SET COLLATE TO "GENERAL" 
? INLIST( "ßt    ", "SST" )        && .T. 
Nicht nur daß das "ß" intern in "ss" umgewandelt wird, Kleinbuchstaben und Großbuchstaben angeglichen werden, es wird sogar das implizite RTRIM() auf den linken String, also auf "ßt    ", durchgeführt. 

Eine andere Funktion, die nach dem gleichen Muster funktioniert, ist BETWEEN(). Hierbei wird die Funktion intern wie ein Vergleich nach folgenden Muster gehandhabt: 

BETWEEN( Var, "String1", "String2" ) 
Var >= "String1" AND var <= "String2" 
Auch hier können wir obiges Beispiel verwenden 

SET EXACT ON 
SET COLLATE TO "GENERAL" 
? BETWEEN( "ßb ", "SSA  ", "SSC" ) && .T. 
Als dritte Funktion im Bunde möchte ich ASCAN() anführen. Intern wird sie durch eine Schleife abgebildet, die einen einfachen Vergleich mit jedem Arrayelement durchführt. Dazu das altbekannte Beispiel: 

DIMENSION aTest[2] 
aTest[1] = "SSA  " 
aTest[2] = "ßb   " 
? ASCAN(atest,"SSB")  && ergibt 2 
Lassen Sie also Vorsicht walten, wenn die Anzahl der Leerzeichen oder die Interpretation der Zeichen von Bedeutung sind. Sollten Sie eine wirklich exakte Suche benötigen, wie sie von "==" durchgeführt wird, so müssen Sie diese Funktionen selbst nachprogrammieren. Besonders riskant ist ihre Verwendung insbesondere dann, wenn sie mit Primärschlüsseln aus binären Zeichen arbeiten. Wenn Sie also ein Array mit Primärschlüsseln auf diese Weise durchsuchen wollen, benutzen Sie dazu unbedingt ihre eigene Routine. Auch Abfragen in einem SELECT-Befehl, der mit BETWEEN() einen Bereich von Primärschlüsseln herausfiltern möchte, sollte lieber umformuliert werden, sonst gelangen unerwünschte Sätze mit in den Ergebniscursor. 

Besonders irritierend ist nun, daß die Funktion AT() und ihre Entsprechung "$" nicht diesen Regeln gehorchen, sondern eine exakte Suche durchführen. So kann es dazu kommen, daß zwar zwei Strings identisch sind, aber dennoch nicht ineinander enthalten sind: 

SET COLLATE TO "GENERAL" 
? "String" = "string"   && .T. 
? "String" $ "string"   && .F. 
Eine besondere Tücke steckt in diesem Befehlen dann, wenn Sie sie auf Tabellenfelder anwenden wollen und mittels SET COLLATE eine deutsche Sortierfolge aktiviert haben. In diesem Fall wird nämlich mit zweierlei Methoden auf die Tabelle zugegriffen. Zugriffe mit "=", ">", "<", BETWEEN(), usw. werden unter Berücksichtigung der Sortierfolge durchgeführt, mit AT() oder "$" dagegen finden sie nur solche Sätze, die auch physikalisch den Suchstring enthalten. Wollen Sie also alle Datensätze finden, in denen das Wort "Straße" auftaucht, und zwar unabhängig davon, ob es "Strasse", "Straße" oder "straße" ist, so müssen Sie zuerst die Groß-/Kleinschreibung anpassen, indem Sie die Strings mit UPPER() umwandeln. Anschließend müssen Sie alle Kombinationen auflisten, die auch SET COLLATE benutzt, in unserem Beispiel also "STRASSE" und "STRAßE": 

BROWSE FOR "STRASSE" $ UPPER(Feld) OR "STRAßE" $ UPPER(Feld) 

Fazit

Wenn immer Sie mit Stringvergleichen arbeiten, machen Sie sich klar, welche der vielen Methoden von FoxPro Sie gerade benötigen und welche eventuell zu falschen Ergebnissen führen. Die folgende Liste enthält einige Checkpunkte, an Hand derer Sie den geeigneten Vergleich finden können: 

Kann der Rest des linken Strings ignoriert werden, wenn der rechte kürzer ist? 

Sind Leerzeichen bei dem Vergleich relevant? 

Handelt es sich um Text oder um binäre Zeichen 

Ist die Groß-/Kleinschreibung egal? 

Stellen Sie in ihren Programmen sicher, daß SET COLLATE und SET EXACT entsprechend dem gewünschten Vergleich gesetzt sind. Ersetzen Sie gegebenenfalls FoxPro-Funktionen durch eigene, wenn die FoxPro-Funktion die falsche Vergleichsmethode verwendet. 

